
Introduction

Questions asked regarding Reliability / ICC

“Advanced techniques are possible for researchers who are interested in providing more information
than a summary statistic”, Hernaez (2015)
Focus: Intraclass correlation (ICC)

most versatile and most potential
Is the classical black box framework the proper way today?
How does Stata support more modern approaches?
Code examples

Intraclass correlation (ICC) / reliability

Definition of agreement, Vet et al. (2006), Hernaez (2015)

Measurement agreement is Measurement variation
How fine can one measure?
A kitchen weight may weight correct within ±5g

The level of non-dectechable variation due to instrument

Intraclass correlation (ICC) / reliability

Definition of reliability / ICC

How well measurements are distinguished despite Measurement variation
1− reliability is the degree of bias due to Measurement variation
A bath weight (correct within ±1kg) is useless in a kitchen

reliability = Variation between study objects
Variation between study objects+Measurement variation , Streiner, Norman, and Cairney (2015)

Variation = Variance ⇒ ANOVA?
ICC ranges from 0 (no reliability) to 1 (perfect reliability)
ICC correlates variables with the same class (unit) and variance, McGraw and Wong (1996)

in contrast to eg Pearsons correlation (example: height(cm) vs weight(kg))

Variation between study objects/Measurement variation = reliability/(1− reliability))
reliability = 0.5⇒ Variation between study objects/Measurement variation = 1
reliability = 0.8⇒ Variation between study objects/Measurement variation = 4
reliability = 0.9⇒ Variation between study objects/Measurement variation = 9

See Koo and Li (2016) for interpretation and reporting

Intraclass correlation (ICC) / reliability

Effect of agreement / Measurement variation on observed

A and B: Same Variation between study objects, different Measurement variation
A and C: same Variation between study objects relative to Measurement variation
B and C: Different Variation between study objects, same Measurement variation
See Dunn (1989) and Vet et al. (2011) on Generalisability theory and reliability

Intraclass correlation (ICC) / reliability

Textbook dataset layout

measurement 1 measurement 2 . . . measurement k
subject 1 y11 y12 . . . y1k
subject 2 y21 y22 . . . y2k

. . . . . . . . . . . . . . .
subject n yn1 yn2 . . . ynk

n subjects (rows) are having k measurements (columns)
Measurements in cells are typically not repeated
Balanced design of single values
Possible bias from measurements (columns)
Shrout and Fleiss (1979) and McGraw and Wong (1996) propose a standardised setup based on
ANOVA

Continuous measurements

Challenges

Is ANOVA the best starting point for ICCs today?

Serious weaknesses of ANOVA estimators, Marchenko (2006)
Possibly negative estimates of variance components
Nonexistence of uniformly best estimators
Lack of uniqueness in the case of unbalanced data

Shrout and Fleiss (1979) and McGraw and Wong (1996) made their suggestion in the early pc years
How to handle ordered or categorical outcomes properly?
Do some measurements needs adjustment?

Example: Measurement precision might dependent on age?

Challenges

Research design and reliability, Zacho et al. (2020)

Four raters from two hospitals using a standard and a new method
Two raters from each hospital

n subjects for each rater
All subjects are rated twice
All raters has used both methods on the n subjects

3 months later a second rating
All subjects are rated once
Standard method is rated within hospital one, new method within hospital two

Outcome has 3 levels:
Benign - 60%
In doubt - 20%
Malignant - 20%

How many research questions are hidden behind this design?
Is a set of pairwise comparisons by ICC (or Kappa) the best way to analyze?

Statistics today

On Anova, maximum likelihood (ML) and restricted maximum likelihood
(REML)

Marchenko (2006) (also see Rabe-Hesketh and Skrondal (2012)):

REML and ML variance estimates are guaranteed to be nonnegative
REML takes into account the implicit degrees of freedom associated with the fixed effects
ANOVA and REML estimators are identical for balanced designs
For unbalanced designs, all three estimators generally differ
ML and REML are preferred methods of estimation for unbalanced data due to simplicity

Statistics today

ICC simplified, Liljequist (2019)

Name Model ICC (agreement)

oneway yij = µ+ Ri + Eij
σ2

R
σ2

R +σ2
E

twoway random yij = µ+ Ri + Ci + Eij
σ2

R
σ2

R +σ2
C +σ2

E

twoway fixed yij = µ+ Ri + ci + Eij
σ2

R
σ2

R +σ̂2
c +σ2

E

Capital letters are random effects
Interaction between subjects and measurements as part of the Error
Same ICC formulas for twoway mixed (pseudo σ̂2c ) and twoway random
Bias over measurements / columns

Agreement or Consistency, see McGraw and Wong (1996) p. 33
Agreement (same level?)
Consistency (Same order?): Leave out bias by measurements σ̂2

c or σ2
C

Do three ICC formulas; oneway; twoway agreement; and twoway consistency

Continuous measurements

Continuous measurements

PEFR example from Rabe-Hesketh and Skrondal (2012) or Bland and
Altman (1986)

17 subjects have their peak expiratory flow rate (PEFR) measured twice with two different instrument
use "http://www.stata-press.com/data/mlmus3/pefr", clear
reshape long wp wm, i(id) j(time)
reshape long w, i(id time) j(pfmeter) string
rename w pefr
strtonum pfmeter
label define pfmeter 1 "mini Wright (l/min)" 2 "Wright (l/min)", replace

Continuous measurements

Using -icc-

you cannot have repeated measurements in twoway -ICC-
icc pefr id pfmeter if time == 1

Intraclass correlations
Two-way random-effects model
Absolute agreement

Random effects: id Number of targets = 17
Random effects: pfmeter Number of raters = 2

--------------------------------------------------------------
pefr | ICC [95% conf. interval]

-----------------------+--------------------------------------
Individual | .9459284 .8574112 .9800787

Average | .972213 .9232325 .9899391
--------------------------------------------------------------
F test that

ICC=0.00: F(16.0, 16.0) = 34.03 Prob > F = 0.000

Note: ICCs estimate correlations between individual measurements
and between average measurements made on the same target.

Continuous measurements

Using -mixed- and -nlcom-

To get same ICCs as from -icc-, the variance components must be crossed
Only one component needs to be crossed, see recipe in Marchenko (2006) and Rabe-Hesketh and
Skrondal (2012)
Confidence intervals not quite the same as for -icc-
For comparison we only look at time 1

mixed pefr if time == 1, reml noheader nolog nofetable ||id: ||_all: R.pfmeter
nlcom ( icc_i: exp(2*_b[lns1_1_1:_cons]) / (exp(2*_b[lns1_1_1:_cons]) ///

+ exp(2*_b[lns2_1_1:_cons]) + exp(2*_b[lnsig_e:_cons])) ), noheader post

------------------------------------------------------------------------------
pefr | Coefficient Std. err. z P>|z| [95% conf. interval]

-------------+----------------------------------------------------------------
icc_i | 0.946 0.026 36.56 0.00 0.895 0.997

------------------------------------------------------------------------------

Continuous measurements

Using -mixed- and -estat icc-

-estat icc- do not work for crossed effects
Described in eg Rabe-Hesketh and Skrondal (2012)
Formula for confidence intervals, see StataCorp LLC (2021 ME) p. 55-56
For comparison we only look at time 1

mixed pefr if time == 1, reml noheader nolog nofetable ||id: ||pfmeter:
estat icc

Intraclass correlation

------------------------------------------------------------------------------
Level | ICC Std. err. [95% conf. interval]

-----------------------------+------------------------------------------------
id | .9460141 .0258752 .8665189 .9792968

pfmeter|id | .980654 3.482216 2.9e-155 1
------------------------------------------------------------------------------

Continuous measurements

Summary, Continuous measurements

Similar ICC estimates
Mixed with crossed variance confidence
interval more similar to traditional ICC
Both mixed-effect models are with option
reml
Several -gsem- attempts with no convergence
(I’m no -gsem- expert)

Continuous measurements

A note on power calculations, Continuous measurements

Lew and Doros (2010) suggests simulations to find optimal for n and k wrt mean width of ICC 95%
CI
Mata and -simulate- makes it easy to simulate the datasets using the kronecker operator (#)
Optimal solution for n and k depends on σsubject , (σmeasurement) and agreement σerror

Example code next slide for the values 1, 0.3 and 0.1 respectively
In this case more subjects is better
More raters is not necessarely better

Alternative is to get the probability of ICC being above a chosen limit, eg 0.8
On next slide

(n,k) = (50, 3) is better than (100, 2)
Precision between 0.01 and 0.03

Continuous measurements

Power simulation, Continuous measurements
capture program drop ciw1
program define ciw1, rclass
1. args n k mu sd_s sd_m err
2. clear
3. mata: out = (1::`n') # J(`k', 1, 1)
4. mata: out = out, rnormal(`n', 1, 0, `sd_s') # J(`k', 1, 1)
5. mata: out = out, J(`n', 1, 1) # (1::`k')
6. mata: out = out, J(`n', 1, 1) # rnormal(`k', 1, 0, `sd_m')
7. mata: out = out, out[., 2] + out[., 4] + rnormal(`n'*`k',1,`mu',`err')
8. mata: nhb_sae_addvars(("s", "m_s", "m", "m_m", "y"), out) // matrixtools
9. mixed y, reml ||s: ||_all:R.m

10. local icc_i_formula exp(2*_b[lns1_1_1:_cons])
11. local icc_i_formula `icc_i_formula' / ( exp(2*_b[lns1_1_1:_cons])
12. local icc_i_formula `icc_i_formula' + exp(2*_b[lns2_1_1:_cons])
13. local icc_i_formula `icc_i_formula' + exp(2*_b[lnsig_e:_cons]) )
14. if `e(converged)' {
15. nlcom ( icc_i: `icc_i_formula'), post
16. lincom _b[icc_i]
17. return scalar ciw = r(ub) - r(lb)
18. }
19. else return scalar ciw = .
20. end
forvalues n = 50(50)150 {
2. forvalues k = 2/4 {
3. quietly simulate ciw = r(ciw), reps(20) nodots: ciw1 `n' `k' 10 2 0.3 0.1
4. quietly g n = `n'
5. quietly g k = `k'
6. quietly if !(`n' == 50 & `k' == 2) append using data/icc1
7. quietly save data/icc1, replace
8. }
9. }
graph dot (mean) ciw, over(k) over(n) name(fig2, replace) ytitle(Mean width of ICC 95% CIs)

Ordered or binary measurements

Ordered or binary measurements

Rating example from StataCorp LLC (2021)

6 subjects (target) are measured by three different raters (judge) using a 1-10 scale (rating)

ordered logistic regression (-meologit-) is often suggested when outcomes are scores
-melogit- for binary measurements

use "https://www.stata-press.com/data/r17/judges", clear

Niels Henrik Bruun (Research data and statistics, Aalborg University Hospital)Regression modelling for Reliability/ICC in Stata 25 / 36



Ordered or binary measurements

Using -icc-

icc rating target judge

Intraclass correlations
Two-way random-effects model
Absolute agreement

Random effects: target Number of targets = 6
Random effects: judge Number of raters = 4

--------------------------------------------------------------
rating | ICC [95% conf. interval]

-----------------------+--------------------------------------
Individual | .2897638 .0187865 .7610844

Average | .6200505 .0711368 .927232
--------------------------------------------------------------
F test that

ICC=0.00: F(5.0, 15.0) = 11.03 Prob > F = 0.000

Note: ICCs estimate correlations between individual measurements
and between average measurements made on the same target.

Ordered or binary measurements

Using -meologit- and -nlcom-

only one component needs to be crossed, see recipe in Marchenko (2006) and Rabe-Hesketh and
Skrondal (2012)
negative lower bound

meologit rating, noheader nolog ||_all: R.judge ||target:
nlcom ( icc_i: _b[var(_cons[target])] / (_b[var(_cons[target])] ///

+ _b[var(_cons[_all>judge])] + _pi^2/3) ), noheader post

------------------------------------------------------------------------------
rating | Coefficient Std. err. z P>|z| [95% conf. interval]

-------------+----------------------------------------------------------------
icc_i | 0.302 0.191 1.58 0.11 -0.073 0.676

------------------------------------------------------------------------------

Ordered or binary measurements

Using -meologit- and -estat icc-

Error variance for a mixed-effects logistic and ordered logistic regression is π2/3, StataCorp LLC
(2021 ME) p. 55
Option intpoint(20) is to achieve convergence

meologit rating, noheader nolog intpoint(20) ||target: ||judge:
estat icc

Residual intraclass correlation

------------------------------------------------------------------------------
Level | ICC Std. err. [95% conf. interval]

-----------------------------+------------------------------------------------
target | .1222287 .2069544 .0031659 .8592619

judge|target | .9349675 .5611143 2.01e-07 1
------------------------------------------------------------------------------

Niels Henrik Bruun (Research data and statistics, Aalborg University Hospital)Regression modelling for Reliability/ICC in Stata 28 / 36



Ordered or binary measurements

Summary, ordered measurements

wide confidence intervals
“meologit,crossed + nlcom” gives similar
estimates to traditional ICC
“meologit,crossed + nlcom” has a negative
lower bound
“meologit,hierachical + estat icc” gives quite
a different estimate
-gsem- not tested

Ordered or binary measurements

A note on power calculations, ordered or binary measurements

Lew and Doros (2010) suggests simulations to find optimal for n and k wrt mean width of ICC 95%
CI
Mata and -simulate- makes it easy to simulate the datasets using the kronecker operator (#)
Inspiration from Buis (2007) and Statalist, Xavier, 2021-05-17
Use code next slide with caution, see Statalist, Enzmann, 2016-06-21
Optimal input should include approximate distribution of the score
Challenge: Interpretation of SDs in the random effects
On next slide precision much lower (between 0.2 and 0.8)

Choose n and k as big as possible
Note (n,k) = (100,3) is better than (150,2)

Ordered or binary measurements

Power simulation, ordered or binary measurements

capture program drop ciw2
program define ciw2, rclass
1. args n k sd_s sd_m percs
2. clear
3. mata: out = (1::`n') # J(`k', 1, 1)
4. mata: out = out, rnormal(`n', 1, 0, `sd_s') # J(`k', 1, 1)
5. mata: out = out, J(`n', 1, 1) # (1::`k')
6. mata: out = out, J(`n', 1, 1) # rnormal(`k', 1, 0, `sd_m')
7. mata: out = out, out[., 2]+out[., 4]+rnormal(`n'*`k',1,0,pi()/sqrt(3))
8. mata: nhb_sae_addvars(("s", "m_s", "m", "m_m", "xb"), out) //matrixtools
9. g rlogit = logit(runiform())

10. _pctile xb, percentiles(`percs')
11. g y = (rlogit > xb + r(r1)) + (rlogit > xb + r(r2))
12. capture meologit y ||_all:R.m ||s:
13. if ! _rc {
14. nlcom ( icc_i: _b[var(_cons[s])] / (_b[var(_cons[s])] /// + _b[var(_cons[_all>m])] + _pi^2/3) ), noheader post
15. lincom _b[icc_i]
16. return scalar ciw = r(ub) - r(lb)
17. }
18. else return scalar ciw = .
19. end
forvalues n = 50(50)150 {
2. forvalues k = 2/4 {
3. quietly simulate ciw = r(ciw), reps(20) nodots: ciw2 `n' `k' 2 0.3 "60 80"
4. quietly g n = `n'
5. quietly g k = `k'
6. quietly if !(`n' == 50 & `k' == 2) append using data/icc2
7. quietly save data/icc2, replace
8. }
9. }
graph dot (mean) ciw, over(k) over(n) name(fig3, replace) ytitle(Mean width of ICC 95% CIs)

Summary

Summary

Take home

From a statistical view, it is better to work modelbased
Model control
Transformations?
Unbalanced datasets
Use of designs
Power (simulation) calculations

On effects (crossed vs hierachical)
StataCorp LLC (2021), and eg Rabe-Hesketh and Skrondal (2012) concentrates on ICC based on
hierachical effects
ICC based on models with crossed effects more similar with ANOVA
In Stata -estat icc- only works with hierachical models

Use -meologit-/-melogit- and sd2
error = π2/3 for ordered/binary categorical variables

Challenge: Interpretation of SDs in the random effects
-gsem- should be appealing - more work required

Summary

Questions?

Thank you!!
References on next slide

Summary

